

NAPREDNI MODELI I BAZE PODATAKA

Upute za projekt iz Semantičkog weba

Ak. godina 2013/2014

Priprema

1. Proučiti slajdove s predavanja i vlastite zabilješke
2. U web pregledniku otvorite url:
<http://www.linkedmdb.org/snorql/>
Dočekati će vas sučelje za interaktivno postavljanje upita nad grafom podataka iz **Linked movie database(LMDB)** izvora.

Alternativno, upite možete postavljati i nad lokalnom kopijom LMDB grafa:

- a) Sa stranica predmeta preuzmite datoteku:
jena-fuseki+LMDBdump.zip
- b) Raspakirajte zip datoteku i iz komandne linije pokrenite Fuseki server:
fuseki-server.bat --loc=LMDB /ds
(proučite i ostale parametre pri pokretanju servera, npr. **--update**)
- c) U web pregledniku otvorite url:
<http://localhost:3030>
na poveznici "control panel" potrebno je izabrati jedini ponuđeni set podataka (dataset) - /ds. Nakon toga otvara se interaktivno sučelje gdje je moguće postavljati upite.

3. Isprobajte nekoliko jednostavnih upita, npr.

- a. **ukupni broj trojki u grafu**

```
SELECT (COUNT(*) AS ?ukupnoTrojki)
WHERE {
    ?s ?p ?o
}
```

- b. **popis svih klasa**

```
SELECT DISTINCT ?class
WHERE {
    [] a ?class
}
ORDER BY ?class
```

- c. **(samo na lokalnoj kopiji!) postoji li u grafu film koji je režirao Quentin Tarantino?**

```
PREFIX movie: <http://data.linkedmdb.org/resource/movie/>
ASK {
    ?movie movie:director ?director .
    ?director movie:director_name "Quentin Tarantino"
}
```

- d. U koliko drama i akcijskih filmova je glumio Will Smith?

```
PREFIX movie: <http://data.linkedmdb.org/resource/movie/>
SELECT (COUNT(DISTINCT ?movie) AS ?brojFilmova)
WHERE {
    ?movie movie:actor ?actor .
    ?actor movie:actor_name "Will Smith" .
    ?movie movie:genre ?genre .
    ?genre movie:film_genre_name ?gname .
    VALUES ?gname {"Action" "Drama"}
}
```

- e. U kojim filmovima, snimljenim nakon 2000. godine, je glumila Catherine Zeta-Jones? Filmove poredati po datumu.

```
PREFIX movie: <http://data.linkedmdb.org/resource/movie/>
PREFIX dc: <http://purl.org/dc/terms/>
SELECT ?title ?date
WHERE {
    ?movie movie:actor ?actor .
    ?movie dc:date ?date .
    ?movie dc:title ?title .
    ?actor movie:actor_name "Catherine Zeta-Jones"
    FILTER(?date>"2000-12-31")
}
ORDER BY ASC(?date)
```

Pokušajte sami napisati nekoliko upita:

- f. Koliko filmova u grafu je režirao Quentin Tarantino?
- g. Koje filmove je režirao Quentin Tarantino?
- h. Tko je sve glumio u filmu „Reservoir Dogs“?
- i. Napišite upit iz d) dijela pomoću ključne riječi UNION
- j. Koje su to drame i akcijski filmovi u kojima je glumio Will Smith?
- k. U kojem filmu u grafu je glumilo najviše glumaca?

Potrebna programska podrška

Za projekt iz Semantičkog weba potrebno je:

JAVA	PHP
<ul style="list-style-type: none">• Apache Jena (apache-jena-2.11.0.zip)• Razvojno okruženje po želji (npr. Eclipse)	<ul style="list-style-type: none">• Arc2 library (arc2-master.zip)• Graphite library (Graphite.php)• Graphite sparqllib (sparqllib.php)

U oba slučaja potreban je Web poslužitelj, za Javu npr. **Apache Tomcat** ili **Jetty**, a ako koristite PHP, npr. **Apache** ili **IIS**.

Alternativno, za rješavanje zadatka možete koristiti i neku drugu kombinaciju tehnologija, no morate biti sigurni da s njima možete uspješno riješiti sve zadatke.

Napomena #1: Ako za vrijeme testiranja upite želite izvršavati nad lokalnom kopijom podataka (npr. zbog brzine ili nedostupnosti mreže/servera i sl.), preuzmite datoteku „**jena-fuseki+LMBDbump.zip**“, koju čini Fuseki server sa „utočenim“ podacima iz LMBDB grafa.

Napomena #2: Za izvršavanje „Federated“ upita potrebna je SPARQL pristupna točka koja „zna“ obaviti raspodjelu upita i udruživanje rezultata, tzv. „mediator endpoint“. Ako koristite **Apache Jena** programski okvir, ugrađeni ARQ engine će obaviti medijaciju. U slučaju upotrebe nekog drugog programskog okvira, za medijaciju možete iskoristiti lokalni **Fuseki** server. U tom slučaju vam ne trebaju podaci, nego samo „čista instalacija“ - **jena-fuseki-1.0.0-distribution.zip**

Napomena #3: Cilj ovog projekta je upoznati se sa tehnologijama Semantičkog weba, naučiti osnove baratanja RDF-om i pisanja SPARQL upita. Zbog toga nije dozvoljeno korištenje tehnologija kao što je LINQ.

Zadatak

Potrebno je ostvariti jednostavan filmski web portal, koji kao izvor podataka koristi semantičke izvore **Linked movie database** i **Dbpedia**. Po želji, može se uključiti i dodatni semantički izvor (npr. Music-brainz, Geonames i sl.). Umjesto **Linked movie database** izvora možete koristiti i lokalni SPARQL endpoint sa kopijom podataka.

Portal treba sadržavati:

- Jednostavnu naslovnicu sa **poljem za pretraživanje po naslovu filma ili imenu glumaca**. Na dnu naslovnice potrebno je ispisati **nekoliko statističkih podataka o izvorima koji se koriste** (npr. ukupni broj trojki, trenutni broj filmova u bazi ili sl. – naravno brojke ne smiju biti „hardcodirane“ nego dohvaćene odgovarajućim upitom) Opcionalno, stranicu možete upotpuniti i drugim sadržajem, npr. s okvirom u kojem se prikazuju osnovne informacije o 3 slučajno odabrana filma, link na stranicu o glumcu koji danas slavi rođendan i sl.
- Stranicu za **prikaz rezultata pretraživanja**. Upit mora pronalaziti sve glumce i filmove koji bilo gdje u imenu ili nazivu sadrže zadani niz znakova. Rezultate je potrebno sortirati abecedno. Rezultati moraju biti prikazani tako da se klikom na odgovarajući link otvara stranica koja prikazuje podatke o tom glumcu ili filmu.
- Stranicu za **prikaz podataka o filmu** (npr. naslov, popis glumaca, datum, redatelj i scenarist, kratki opis...). Potrebno je prikazati najmanje 5 različitih podataka. Dohvat podataka potrebno je obaviti **bez** SPARQL upita (korištenjem metoda za manipulaciju trojkama u RDF-u). **Obavezno** dohvatiti i podatke sa Dbpedije (ako u LMDB za taj film postoji owl:sameAs URI prema Dbpediji)
- Stranicu za **prikaz podataka o jednom glumcu** (npr. ime i prezime, datum i mjesto rođenja, fotografija, wiki stranica, kratki životopis...) . Podatke o glumcima potrebno je dohvatiti kroz združeni (en. *federated*) SPARQL upit. Potrebno je prikazati najmanje 5 različitih podataka. Ako ne postoji owl:sameAs link na Dbpediju, pokušajte napraviti združeni upit preko imena glumca. Uočite da LMDB nema oznaku jezika, dok Dbpedija ima (hint: `FILTER (STR(?imeDBP) = ?imeLMDB)`).
- Potrebno je omogućiti **dohvat podataka o filmovima i glumcima na različitim jezicima**. Jezik se može izabrati npr. preko padajućeg izbornika na samoj stranici.

Linked movie database

Linked movie database je otvorena baza filmova, povezana s raznim drugim semantičkim izvorima, kao što su Dbpedia, Geonames i sl.

WEB: <http://www.linkedmdb.org/>

SPARQL endpoint: <http://data.linkedmdb.org/sparql>

HTML verzija: npr. <http://data.linkedmdb.org/page/film/2014>

LOKALNA KOPIJA PODATAKA, interaktivno sučelje: <http://localhost:3030/>

LOKALNA KOPIJA PODATAKA, SPARQL endpoint: <http://localhost:3030/ds/sparql>

Dbpedia

Dbpedia je, prema definiciji s njihovih web stranica, „community effort“ tj. otvoreni zajednički projekt izdvajanja strukturiranih informacija iz Wikipedije. Izdvojeni podaci dostupni su u formi RDF grafa na Webu i omogućavaju postavljanje kompliciranih upita i pretraživanje Wikipedije na nove i prije Dbpedije nemoguće načine.

WEB: <http://dbpedia.org/>

SPARQL endpoint: <http://dbpedia.org/sparql>

HTML verzija: npr. [http://dbpedia.org/page/The Lord of the Rings](http://dbpedia.org/page/The_Lord_of_the_Rings)

Primjeri SPARQL upita s predavanja

LMDB: Upit koji pronalazi sve filmove u kojima je glumio ili ih režirao Clint Eastwood

```
PREFIX lmdb: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?movie
WHERE {
    {
        ?movie lmdb:actor ?personURI .
        ?personURI lmdb:actor_name "Clint Eastwood"
    }
    UNION
    {
        ?movie lmdb:director ?personURI .
        ?personURI lmdb:director_name "Clint Eastwood"
    }
}
```

LMDB: Popis svih glumaca koji su glumili sa Kevinom Baconom

```
PREFIX lmdb: <http://data.linkedmdb.org/resource/movie/>
SELECT DISTINCT ?actorName
WHERE {
    ?kb lmdb:actor_name "Kevin Bacon" .
    ?movie lmdb:actor ?kb .
    ?movie lmdb:actor ?actor .
    ?actor lmdb:actor_name ?actorName .
    FILTER (?kb != ?actor)
}
ORDER BY ASC(?actorName)
```

OPĆENITO: Upit koji računa ukupni broj trojki u grafu

```
SELECT (COUNT(*) AS ?ukupnoTrojki)
WHERE {
    ?s ?p ?o
}
```

DBPEDIA: Upit koji ispisuje listu svih gradova u Hrvatskoj s engleskim nazivima i opisima

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
SELECT ?grad ?ime ?kratkiOpis
WHERE {
    ?grad a dbpedia-owl:City.
    ?grad dbpedia-owl:country <http://dbpedia.org/resource/Croatia> .
    ?grad rdfs:label ?ime .
    ?grad dbpedia-owl:abstract ?kratkiOpis
    FILTER(lang(?ime) = "en")
    FILTER(lang(?kratkiOpis) = "en")
}
```

LMDB+DBPEDIA: Upit koji se vraća podatke o filmovima Arnolda Schwarzeneggera

```
PREFIX lmbd: <http://data.linkedmdb.org/resource/movie/>
PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX dbpo: <http://dbpedia.org/ontology/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
SELECT ?dbpediaUrl ?title ?date ?abstract ?runtime ?wikiPage ?picURL
WHERE {
    ?actor lmbd:actor_name "Arnold Schwarzenegger".
    ?movie lmbd:actor ?actor .
    ?movie dcterms:title ?title .
    ?movie dcterms:date ?date .
    ?movie owl:sameAs ?dbpediaUrl .
    FILTER( REGEX( STR(?dbpediaUrl), "dbpedia")) .
    SERVICE <http://dbpedia.org/sparql>
    {
        ?dbpediaUrl a dbpo:Film .
        ?dbpediaUrl foaf:isPrimaryTopicOf ?wikiPage .
        ?dbpediaUrl dbpo:abstract ?abstract .
        ?dbpediaUrl dbpo:runtime ?runtime .
        ?dbpediaUrl foaf:depiction ?picURL .
        FILTER(lang(?abstract) = "en")
    }
}
```

Poveznice na dodatnu literaturu, dokumentaciju i sl.

Apache Jena i Fuseki server

<http://jena.apache.org/>

Graphite PHP Linked Data Library i sparqlLib.php

<http://graphite.ecs.soton.ac.uk/>